





Gradle: Réalisez vos builds avec Gradle

David THIBAU – 2020

david.thibau@gmail.com



Agenda

- Introduction

- Attentes sur les outils de build
- Gradle vs Ant/Maven
- Installation, commandes
- Un bref tour de Groovy

- Concepts coeur

- Composants d'un projet
- Manipulation des tâches
- Hook du cycle de vie
- Plugins, le plugin init, le wrapper

- Java et C++

- Gestion de Dépendances
- Plugins pour Java
- Plugins pour C++

- Multi-projets

- Multi-projets

- Livraison continue

- Tests
- Analyse de code
- Jenkins



Introduction

Attentes sur les outils de build
Positionnement vis à vis de Ant, Maven
Installation, Commandes
Un bref tour de Groovy



Pré-requis d'un build


- ♦ **Proscrire les interventions manuelles** sujettes à erreur et chronophage
- ♦ Créer des builds **reproductibles** : Pour tout le monde qui exécute le build
- ♦ **Portable** : Ne doit pas nécessiter un OS ou un IDE particulier, il doit être exécutable en ligne de commande
- ♦ **Sûr** : Confiance dans son exécution



Graphe de build

Un build est une séquence ordonnée de tâches unitaires.

Les tâches ont des dépendances entre elles qui peuvent être modélisées via un **graphe acyclique dirigé** :





Composants d'un outil de build

Le fichier de build : Contient la configuration requises pour le build, les dépendances externes, les instructions pour exécuter un objectif sous forme de tâches inter-dépendantes

Une tâche prend une entrée effectue des traitements et produit une sortie. Une tâche dépendante prend comme entrée la sortie d'une autre tâche

Moteur de build : Le moteur traite le fichier de build et construit sa représentation interne. Des outils permettent d'accéder à ce modèle via une API

Gestionnaire de dépendances : Traite les déclarations de dépendances et les résout par rapport à un dépôt d'artefact contenant des méta-données permettant de trouver les dépendances transitive



Ant

- **Apache Ant (Another Neat Tool)** : L'objectif principal est de d'automatiser les tâches typiques nécessaires à un projet Java (Compilation, Tests unitaires, Packaging Jar, Javadoc, ...)
- Fournit de nombreuses tâches prédéfinies et peut être étendu avec de nouvelles tâches écrites en Java
- Ne fournit pas de gestionnaire de dépendances mais s'intègre facilement avec Ivy
- Fichier de build en XML (*build.xml*)



Example

```
<project name="my-app" default="dist" basedir=". ">
  <property name="src" location="src"/>
  <property name="build" location="build"/>
  <property name="dist" location="dist"/>
  <property name="version" value="1.0"/>


  <target name="init">
    <mkdir dir="${build}"/>
  </target>

  <target name="compile" depends="init" description="compile the source">
    <javac srcdir="${src}" destdir="${build}"/>
  </target>

  <target name="dist" depends="compile" description="generate the distribution">
    <mkdir dir="${dist}"/>
    <jar jarfile="${dist}/my-app-${version}.jar" basedir="${build}"/>
  </target>

  <target name="clean" description="clean up">
    <delete dir="${build}"/>
    <delete dir="${dist}"/>
  </target>
</project>
```

Example






Inconvénients

- Usage de XML => Fichier de build long
- Logique de build complexe conduit à des scripts de builds longs et difficilement maintenable (*if-then-else* avec un langage de markup!!).
- *Ant* ne donne pas de recommandation pour la mise en place d'un projet => chaque projet est différent, bcp de copier/coller entre projets. Chaque nouveau développeur doit comprendre le build du projet.
- N'expose pas d'API permettant de connaître le modèle interne ni de métriques sur l'exécution.
- *Ant* sans *Ivy* nécessite une gestion manuelle des dépendances
- *Ant* avec *Ivy* nécessite un effort d'intégration supplémentaire




Maven


- Maven applique l'idée « Convention plutôt que configuration »
=> un projet complet si il adhère aux conventions nécessite peu de lignes de XML
- Maven permet de générer de la documentation (Javadocs)
- Les fonctionnalités cœur peuvent être étendues grâce à des plugins. Les plugins sont nombreux, développer son propre plugin n'est pas anodin.
- Maven définit le cycle de vie d'un build comme une suite ordonnée de phase. Chaque phase correspond à l'exécution de plugins prédéfinis ou custom
- Maven gère les dépendances et propose des dépôts d'artefacts
- Il gère également les projets multi-modules



Phases Maven



Dépendances et dépôts





Exemple *pom.xml*

```
<project xmlns="http://maven.apache.org/POM/4.0.0"
  xmlns:xsi="http://www.w3.org/2001/XMLSchema-instance"
  xsi:schemaLocation="http://maven.apache.org/POM/4.0.0
    http://maven.apache.org/xsd/maven-4.0.0.xsd">

  <modelVersion>4.0.0</modelVersion>
  <groupId>com.mycompany.app</groupId>
  <artifactId>my-app</artifactId>
  <packaging>jar</packaging>
  <version>1.0</version>

  <dependencies>
    <dependency>
      <groupId>org.apache.commons</groupId>
      <artifactId>commons-lang3</artifactId>
      <version>3.1</version>
      <scope>compile</scope>
    </dependency>
  </dependencies>
</project>
```



Inconvénients

- Comme avec Ant, le format XML rend le fichier *pom.xml* long et + difficile à lire et appréhender
- La structure par défaut et le cycle de vie imposé sont souvent trop restrictifs et peuvent ne pas correspondre aux besoins d'un projet
- Le développement de plugins est assez lourd : Connaître les Mojos, fournir une description en XML et annotations Java
- Les versions d'avant 2.0.9 essayaient d'automatiquement mettre à jour les plugins cœur, ce qui provoquait de l'instabilité.
- Une dépendance téléchargée dans le dépôt local ne stocke pas sa provenance. Cela peut être un souci de portabilité



Objectifs de Gradle

- Un langage de build expressif, déclaratif et maintenable
- Des arborescences et cycle de vie projet standardisés, mais offrant une flexibilité complète
- Possibilité de facilement ajouter de la logique personnalisée
- Support pour les multi-projets
- Support pour la gestion de dépendances.
- Facilité d'intégration et de migration des outils existants (Ant/Maven).
- Builds performants et scalable



Contexte

Les projets actuels utilisent des piles logicielles variées faisant usage de différents langages de programmation, de framework de test ou d'analyse.

Avec l'arrivée des méthodes agiles, le build doit permettre l'intégration continue de nouveau code et faciliter les livraisons dans les environnements de test et de production.



Code plutôt que XML


En gardant une approche configuration par convention, *Gradle* permet d'implémenter son build de façon déclarative en utilisant un DSL implémenté en *Groovy* ou *Kotlin*

- De la logique personnalisée peut alors être codée en *Groovy* ou *Kotlin*

Gradle fournit également sa propre implémentation pour la gestion des dépendances et les projets multi-modules ; tout en essayant le plus possible d'être compatible avec Ivy et Maven

Gradle aide à la migration en supportant directement les tâches Ant

Correspondance fichier de build et modèle interne





Build is code

Chaque élément d'un script *Gradle* correspond à une classe Java.

- Groovy/Kotlin permettent juste de rendre plus compact l'écriture de code avec la programmation fonctionnelle
- Gradle expose également des hooks dans les phases de son cycle de vie, permettant de configurer, monitorer, influencer l'exécution




Conventions ... flexibles

Gradle fournit des recommandations et des valeurs par défaut sensées pour les projets.

- Les tâches par défaut suffisent dans pas mal de cas.
- Mais, *Gradle* permet également de facilement remettre en cause ses conventions

Gradle fournit des archétypes pour les projets *Java*, *Scala*, *Groovy*, *Kotlin*, *C++*.

Java





Scalable

Gradle, pour des soucis de scalabilité et de performance

- supporte les builds incrémentaux en gérant les entrées sorties des différentes tâches du build
- Supporte l'exécution // des tests
- Utilise un démon, pour optimiser les performances des build



OpenSource

Gradle est sous licence Apache License 2.0

1ère release Avril 2008

Forum : <https://discuss.gradle.org/>

Offre commerciale et support :
Gradleware



Installation



Moyens

Gradle peut être installé sur Linux, MacOS ou Windows

- via des gestionnaires de paquet comme SDKMAN, Homebrew, ou Scoop
- Ou manuellement

Pour les mises à jour, utiliser Gradle Wrapper.



Pré-requis

JDK 7+

Java est dans le Path ou indiqué dans
JAVA_HOME

Groovy est livré avec la distribution de
Gradle



Gestionnaire de paquets

Linux : SDK permet l'installation de plusieurs versions

```
sdk install gradle 4.6
```

MacOS : Homebrew et MacPort

```
brew install gradle
```

```
sudo port install gradle
```

Windows :

```
scoop install gradle
```

```
choco install gradle
```



Installation manuelle

Téléchargement de la distribution
(binaires ou sources)

Dézipper

Ajouter `${GRADLE_HOME}/bin` dans le
PATH

=> Vérification :
`gradle -v`



Premier script

La commande **gradle** recherche le fichier nommé par défaut **build.gradle** dans son répertoire

Le fichier définit des **tâches**

Exemple *HelloWorld* :

```
task helloWorld {  
    doLast {  
        println 'Hello world!'  
    }  
}  
---  
gradle -q helloWorld
```



Autre exemple

```
// L'opérateur << est un raccourci pour doLast
```

```
task startSession << {  
    chant()  
}
```

```
// Utilisation d'une tâche Ant
```

```
def chant() {  
    ant.echo(message: 'Repeat after me...')  
}
```

```
// Définition dynamique de tâches
```

```
3.times {  
    task "yayGradle${it}" << {  
        println 'Gradle rocks'  
    }  
}
```

```
// Dépendances entre tâches
```

```
yayGradle0.dependsOn startSession  
yayGradle2.dependsOn yayGradle1, yayGradle0  
task groupTherapy(dependsOn: yayGradle2)
```




Commande en ligne

gradle tasks : Liste les tâches disponibles d'un projet groupées par groupe de tâches

gradle properties : Les propriétés du projet disponibles

Abréviation en camelCase : *gradle gT*

Option ***-x*** pour exclure une tâche



Options

Principales options :

- ?, -h, --help**: Affiche toutes les options disponibles
- b, --build-file**: Le fichier de build, par défaut *build.gradle*
- offline**: Mode offline pour tester si le cache local a tout ce qu'il faut
- D, --system-prop**: Propriété JVM
- P, --project-prop** : Propriété du projet
- i, --info**: Logger Gradle à INFO
- s, --stacktrace**: Affichage de la stack trace en cas d'erreur
- q, --quiet**: Quiet mode

Les options peuvent être combinées, exemple -is



Démon Gradle

Un démon *gradle* démarre automatiquement après la première commande *gradle*.

- Il est utilisé par les builds suivants afin d'optimiser les temps de build.
- Il s'arrête automatiquement après 3h d'inactivité



Un bref tour de Groovy



Sources

Les fichiers sources de Groovy sont soit :

- Des définitions de classe
- Des scripts qui pourront alors être démarrés par une commande en ligne.



Déclaration de classes

La déclaration de classe est similaire à Java, les méthodes sont *public* par défaut

```
class Book {  
    private String title  
    Book (String theTitle) {  
        title = theTitle  
    }  
    String getTitle(){  
        return title  
    }  
}
```



Scripts

Les **scripts** sont des fichiers texte (ayant par convention l'extension **.groovy*)

Il peuvent être exécutés par :

> `groovy myfile.groovy`

- Les scripts contiennent des instructions qui ne sont pas encapsulées par une déclaration de classe.
- Ils peuvent contenir des définitions de méthodes
- Un script est parsé intégralement avant son exécution



Exemple de script

Book.groovy est dans le classpath

```
Book gina = new Book('Groovy in Action')  
assert gina.getTitle() == 'Groovy in Action'
```

Appel de méthode avant sa déclaration

```
assert getTitleBackwards(gina) == 'noitcA ni yvoorG'  
String getTitleBackwards(book) {  
    String title = book.getTitle()  
    return title.reverse()  
}
```




GroovyBeans

Un **GroovyBean** est un JavaBean défini en Groovy.

Groovy facilite le travail avec des beans :

- Il génère les accesseurs
- Il permet un accès simplifié à tous les JavaBeans (et les GroovyBeans)
- Il simplifie l'enregistrement des gestionnaires d'événements associés

```
class BookBean {  
    String title  
}  
def groovyBook = new BookBean()  
// Appel du setter  
groovyBook.title = 'Groovy conquers the world'
```



Annotations

Groovy permet d'utiliser et définir des **annotations** comme Java

- Exemple : *@Immutable* qui est sensiblement équivalent à *final*

Les annotations *Groovy* instruisent le compilateur afin que le compilateur effectue une transformation AST (abstract syntax tree), i.e. ajout/suppression de méthodes, modification de la structure de code.



@Grab

L'annotation **@Grab** est utilisée pour définir explicitement les dépendances vers des librairies externes.

A la compilation et l'exécution, la librairie est téléchargée si besoin (*.groovy/grape*) et ajoutée au classpath.

```
@Grab('commons-lang:commons-lang:2.4')
import org.apache.commons.lang.ClassUtils
class Outer {
    class Inner {}
}
assert !ClassUtils.isInnerClass(Outer)
assert ClassUtils.isInnerClass(Outer.Inner)
```



String

En Groovy, les littéraux String peuvent utiliser les simples ou double-quotes.

La version double-quotes permet l'utilisation d'expressions qui sont résolues à l'exécution

```
def nick = 'ReGina'  
def book = 'Groovy in Action, 2nd ed.'  
assert "$nick is $book" == 'ReGina is Groovy in Action, 2nd  
ed.'
```



Pas de types primitifs

En Groovy, tout est objet.

- Tout nombre, booléen est converti en une objet Java.
- Les notations des types primitifs sont cependant toujours supportées :

```
def x = 1
def y = 2
assert x + y == 3
assert x.plus(y) == 3
assert x instanceof Integer
```



Collections

Groovy facilite la manipulation des collections en ajoutant des opérateurs, des instanciations via des littéraux et de nouvelles méthodes

Il introduit également un nouveau type :
Range



Exemples

```
// roman est une List
def roman = ['', 'I', 'II', 'III', 'IV', 'V', 'VI', 'VII']
// On accède à un élément comme un tableau Java
assert roman[4] == 'IV'
// Il n'y a pas d'ArrayIndexOutOfBoundsException
roman[8] = 'VIII'
assert roman.size() == 9
// Les maps peuvent être facilement instanciées
def http = [
  100 : 'CONTINUE',
  200 : 'OK',
  400 : 'BAD REQUEST'
]
// Accès aux éléments avec la notation Array Java
assert http[200] == 'OK'
// Méthode put simplifiée
http[500] = 'INTERNAL SERVER ERROR'
assert http.size() == 4
```



Closures

Les ***closures*** Groovy permettent la programmation fonctionnelle.

Un bloc d'instructions peut être passé en paramètre à une méthode.

Un objet de type *Closure* travaille en sous-main

```
[1, 2, 3].each { entry -> println entry }
```




Structures de contrôle

```
if (false) assert false // if sur une ligne
```

```
if (null) { // null est false
```

```
    assert false
```

```
}
```

```
// Boucle while classique
```

```
def i = 0
```

```
while (i < 10) {
```

```
    i++
```

```
}
```

```
assert i == 10
```

```
// for in sur un intervalle
```

```
def clicks = 0
```

```
for (remainingGuests in 0..9) {
```

```
    clicks += remainingGuests
```

```
}
```

```
assert clicks == (10*9)/2
```



Structures de contrôle (2)

// for in sur une liste

```
def list = [0, 1, 2, 3]
for (j in list) {
    assert j == list[j]
}
```

// each avec une closure

```
list.each() { item ->
    assert item == list[item]
}
```

// Switch

```
switch(3) {
    case 1 : assert false; break
    case 3 : assert true; break
    default: assert false
}
```



Types de données et collection



Tout Objet

Dans *Groovy*, tout est objet. Il n'y a pas de type primitif comme en Java

- *Groovy* inter-opère avec Java via du boxing et du unboxing automatique

Groovy permet de spécifier explicitement le type d'une variable **ou** de l'omettre

Le mot-clé ***def*** est utilisé pour indiquer qu'aucun type n'est spécifié.



Type safe

Même si le type de variable peut être implicite, Groovy est ***type safe*** :
Un objet d'un type particulier ne peut pas être assigné à un autre type si il n'y a pas de conversion définie.

=> On peut donc omettre les marqueurs de type mais Groovy effectuera les vérifications de type à l'exécution.



Surcharge d'opérateurs

Groovy base ses opérateurs sur des appels de méthode.

- Ainsi, les opérateurs peuvent être surchargés et s'appliquer à différents types
- Les opérateurs peuvent alors être utilisés sur nos propres classes du moment qu'elles implémentent la méthode associée (Encore une fois, pas besoin d'implémenter une interface comme en Java !).



Examples

Operator	Name	Method	Works with
<code>a + b</code>	Plus	<code>a.plus(b)</code>	Number, String, StringBuffer, Collection, Map, Date, Duration
<code>a - b</code>	Minus	<code>a.minus(b)</code>	Number, String, List, Set, Date, Duration
<code>a * b</code>	Star	<code>a.multiply(b)</code>	Number, String, Collection
<code>a / b</code>	Divide	<code>a.div(b)</code>	Number
<code>a % b</code>	Modulo	<code>a.mod(b)</code>	Integral number
<code>a++</code> <code>++a</code>	Postincrement Preincrement	<code>def v = a; a = a.next(); v</code> <code>a = a.next(); a</code>	Iterator, Number, String, Date, Range
<code>a--</code> <code>--a</code>	Postdecrement Predecrement	<code>def v = a; a = a.previous(); v</code> <code>a = a.previous(); a</code>	Iterator, Number, String, Date, Range
<code>-a</code>	Unary minus	<code>a.unaryMinus()</code>	Number, ArrayList



String

Les chaînes de caractères Groovy peuvent être :

- ***java.lang.String***
- ***groovy.lang.Gstring***

Les *GStrings* permettent de spécifier des emplacements (*placeholder*) pouvant être évalués à l'exécution (i.e. interpolation de *String*)



Littéraux

Groovy permet différentes options pour les littéraux :

- Les **simple quotes** pour les *String* Java
- Les **double-quotes** sont équivalentes au simple quote sauf si le texte contient un **\$** (placeholder). Dans ce cas, il s'agit d'une *Gstring*.
- Les **triples-quotes** (**simple** ou **double**) permettent du texte sur plusieurs lignes . Elles peuvent être de simples *String* ou des *GString*
- Si le texte démarre par un **/** ou un **\$/** (légères différences), Il n'est pas nécessaire d'échapper les backslash et cela facilite l'écriture de *regex*



Examples

Start/end characters	Example	Placeholder resolved?	Backslash escapes?
Single quote	<code>'hello Dierk '</code>	No	Yes
Double quote	<code>"hello \$name "</code>	Yes	Yes
Triple single quote (' ' ')	<code>'''===== Total: \$0.02 ====='''</code>	No	Yes
Triple double quote (" " ")	<code>"""first \$line second \$line third \$line"""</code>	Yes	Yes
Forward slash	<code>/x(\d*)y/</code>	Yes	Occasionally
Dollar slash	<code>\$/x(\d*)y/\$</code>	Yes	Occasionally



GString

Une *GString* contient des placeholders exprimés par ***$\${expression}$*** ou tout simplement ***$\$reference$*** .

```
def me = 'Tarzan'  
def you = 'Jane'  
def line = "me $me - you $you"  
assert line == 'me Tarzan - you Jane'
```



Méthodes additionnelles

```
print 'Hello Groovy!'
String greeting = 'Hello Groovy!'
// Utilisation de range
assert greeting[6..11] == 'Groovy'
// Opérateur -
assert 'Hi' + greeting - 'Hello' == 'Hi Groovy!'
// Comptage de caractère
assert greeting.count('o') == 3
// Ajout de caractère à droite ou gauche, centrage
assert 'x'.padLeft(3) == ' x'
assert 'x'.padRight(3, '_') == 'x__'
assert 'x'.center(3) == ' x '
// Opérateur *
assert 'x' * 3 == 'xxx'
```



regex

Groovy s'appuie sur Java pour les expressions régulières et ajoute 3 opérateurs :

- find , *=~*
- match , *==~*
- pattern , *~string*

L'écriture de *regex* est facilitée via le marqueur de String */*

```
assert "\\d" == /\d/
```



Exemples

boucle sur les occurrence


```
def myFairStringy = 'The rain in Spain stays mainly in the plain!'
def wordEnding = /\w*ain/
def rhyme = /\b$wordEnding\b/
def found = ''
myFairStringy.eachMatch(rhyme) { match ->
  found += match + ' '
}
assert found == 'rain Spain plain '
found = ''
(myFairStringy =~ rhyme).each { match ->
  found += match + ' '
}
assert found == 'rain Spain plain '
```



Règles de conversion des nombres

Pour les opérations + , - et *

- Si chaque opérande est *Float* ou un *Double* , le résultat est *Double*. (A la différence de Java qui si les 2 opérandes sont des *Float* le résultat est *Float*)
- Sinon, si un des opérandes est *BigDecimal* , le résultat est un *BigDecimal*
- Sinon, si un des opérandes est *BigInteger* , le résultat est un *BigInteger*
- Sinon, si un des opérandes est *Long*, le résultat est *Long*
- Sinon le résultat est *Integer*



Méthodes additionnelles sur les nombres

```
def store = ''  
// Répétition  
10.times{ store += 'x' }  
assert store == 'xxxxxxxxxx'  
// Parcours  
store = ''  
1.upto(5) { number -> store += number }  
assert store == '12345'  
  
store = ''  
2.downto(-2) { number -> store += number + ' '}  
assert store == '2 1 0 -1 -2 '  
// Parcours avec un pas  
store = ''  
0.step(0.5, 0.1){ number -> store += number + ' '}  
assert store == '0 0.1 0.2 0.3 0.4 '
```





Collections

Groovy facilite l'utilisation des collections

- Les listes, maps et intervalles sont de simples types de données et peuvent être déclarés par des littéraux.
- Ils ont des opérateurs spécialisés et de nombreuses améliorations par rapport au JDK

La notation utilisée est différente de Java mais la sémantique reste la même

Une collection n'est pas disponible en Java : les intervalles (ranges)



Range

Une collection de type **Range** a une limite inférieure et supérieure ainsi qu'une stratégie de parcours

Elle permet :

- d'itérer sur chaque élément (*each*)
- De déterminer si un élément appartient à l'intervalle


Les éléments peuvent être de n'importe quel type du moment qu'ils :

- Implémentent les méthodes *next* et *previous* (surcharge des opérateurs ++ et --).
- Implémentent *java.lang.Comparable* ; méthode *compareTo* surchargeant l'opérateur <=>



Exemples (1)

```
assert (0..10).contains(0)
assert (0..10).contains(10)
assert (0..<10).contains(10) == false
// Références à un intervalle
def a = 0..10
a = new IntRange(0,10)
// Intersection d'intervalle
assert (0.0..1.0).contains(1.0)
// Dates
def today = new Date()
def yesterday = today - 1
assert (yesterday..today).size() == 2
// Caractère
assert ('a'..'c').contains('b')
```



Exemples (2)

// Boucle for in

```
def log = ''  
for (element in 5..9){ log += element }  
assert log == '56789'  
log = ''  
for (element in 9..5){ log += element }  
assert log == '98765'
```

// Méthode each

```
log = ''  
(9..<5).each { element -> log += element }  
assert log == '9876'
```



Listes

Les listes Groovy sont par défaut de type ***java.util.ArrayList***

Elles peuvent également être déclarées explicitement avec un constructeur approprié (ex : *new LinkedList(myList)*)

- Les listes peuvent être déclarées et initialisés avec

```
List myList = [1, 2, 3]
```

- Les éléments peuvent être accédés via

```
MyList[0]
```

- Un intervalle peut être converti en List

```
List longList = (0..1000).toList()
```

GDK étend tous les tableaux, les collections et les strings la méthode *toList* qui retourne une nouvelle List.

Il ajoute des méthodes à l'interface *List*, *Collection* ... et *Object*



Opérateurs

GDK surcharge la méthode ***getAt*** en acceptant un argument de type pour accéder à un sous-ensemble de la collection.

La même stratégie est appliquée à la méthode ***putAt***

```
myList = ['a','b','c','d','e','f']
assert myList[0..2] == ['a','b','c']
assert myList[0,2,4] == ['a','c','e']
myList[0..2] = ['x','y','z']
assert myList == ['x','y','z','d','e','f']
myList[3..5] = []
assert myList == ['x','y','z']
myList[1..1] = [0, 1, 2]
assert myList == ['x', 0, 1, 2, 'z']
```



Opérateurs (2)

Les opérateurs *plus(Object)* , *plus(Collection)* , *leftShift(Object)* , *minus(Collection)* , et *multiply* sont supportés par les listes

```
myList = []  
myList += 'a'  
assert myList == ['a']  
myList += ['b', 'c']  
assert myList == ['a', 'b', 'c']  
myList = []  
myList << 'a' << 'b'  
assert myList == ['a', 'b']  
assert myList - ['b'] == ['a']  
assert myList * 2 == ['a', 'b', 'a', 'b']
```



Structure de contrôle

Les listes peuvent être utilisées dans des *if* , *switch* et *for*

```
myList = ['a', 'b', 'c']
// Switch value
assert myList.isCase('a')
assert 'b' in myList
def candidate = 'c'
switch(candidate){
  case myList : assert true; break
  default : assert false
}
assert ['x','a','z'].grep(myList) == ['a']
myList = []
if (myList) assert false
def expr = ''
for (i in [1,'*',5]){
  expr += i
}
assert expr == '1*5'
```




Map (1)

Les Maps sont par défaut de type *java.util.LinkedHashMap*.

Elles peuvent également être déclarées explicitement en appelant un constructeur approprié

Elles peuvent être initialisées comme suit :

```
[key1:value1, key2:value2, key3:value3]
```



Map (2)

Les maps permettent de récupérer ou de mettre à jour des valeurs en indiquant la clé :

```
def myMap = [a:1, b:2, c:3]
assert myMap['a']== 1
def emptyMap = [:]
assert emptyMap.size() == 0
```

L'opérateur de propagation permet de référencer toutes les valeurs d'une map

```
def myMap = [a:1, b:2, c:3]
def composed = [x:'y', *:myMap]
assert composed == [x:'y', a:1, b:2, c:3]
```



Exemples

```
def myMap = [a:1, b:2, c:3]
// Récupérer un élément existant
assert myMap['a'] == 1
assert myMap.a == 1
assert myMap.get('a') == 1
assert myMap.get('a',0) == 1
// Essayer de récupérer un élément manquant
assert myMap['d'] == null
assert myMap.d == null
assert myMap.get('d') == null
// Valeur par défaut
assert myMap.get('d',0) == 0
assert myMap.d == 0
// Mise à jour
myMap['d'] = 1
assert myMap.d == 1
myMap.d = 2
assert myMap.d == 2
```



Méthodes GDK

GDK ajoute les méthodes ***any*** et ***every*** qui retournent un *Boolean* indiquant si une ou toute les entrées de la Map satisfont une closure donnée.

```
assert myMap.any {entry -> entry.value > 2 }  
assert myMap.every {entry -> entry.key< 'd'}
```



Boucles

Il est possible d'itérer sur les entrées ou sur les clés et valeurs séparément

```
def myMap = [a:1, b:2, c:3]
// Boucle sur les entrées
def store = ''
myMap.each { entry ->
store += entry.key
store += entry.value
}
assert store == 'a1b2c3'
// Boucle sur les clés et valeurs
store = ''
myMap.each {store += key ; store += value}
assert store== 'a1b2c3'
// Boucle sur les clés
store = ''
for (key in myMap.keySet()) {store += key}
assert store == 'abc'
// Boucle sur les valeurs
store = ''
for (value in myMap.values()) {store += value}
assert store == '123'
```



Concepts cœur

Composants d'un build
Manipulation des tâches
Hook du cycle de build
Plugins



Composants d'un build

Chaque build Gradle est constitué de **projets**, de **tâches** et de **propriétés**.

Chaque build contient au moins 1 projet qui contient une ou plusieurs tâches.

Les projets et les tâches exposent des propriétés qui peuvent être utilisées pour contrôler le build

Les projets et les tâches correspondent à des classes de l'API Gradle



Projet

Lors du démarrage d'un build, *Gradle* instancie une classe ***org.gradle.api.Project*** en fonction du script *build.gradle* et rend la variable ***project*** disponible

C'est la variable implicite

```
// Utilisation de la variable implicite
setDescription("myProject")
println "Description of project $name: " +
    project.description
```




Initialisation du build

Durant l'initialisation, Gradle assemble un objet *Project* pour chaque fichier "*build.gradle*" participant au build comme suit :

- Crée une instance de la classe ***Settings***
- Évalue le script "***settings.gradle***" si il est présent afin de configurer l'instance *Settings*
- Utilise l'objet *Settings* pour créer la hiérarchie des instances projets
- Finalement, évalue chaque projet en exécutant son fichier "***build.gradle***".


Un projet parent est évalué avant ses enfants.

L'ordre peut être contrôlé

- explicitement par des méthodes (*evaluationDependsOnChildren()* ou *evaluationDependsOn(String)*)
- ou implicitement par les dépendances projet



Interface *Project*





Tâches

Une **tâche** représente une partie atomique d'un build comme la compilation ou la génération de la javadoc.

Chaque tâche appartient à un projet.

Il est possible d'utiliser les méthodes de **TaskContainer** pour créer ou rechercher des instances de tâche

- Par ex : *TaskContainer.create(String)* crée une tâche vide avec un nom.

Le mot-clé **task** peut également être utilisé dans le fichier de build :

```
task myTask
task myTask { configure closure }
task myTask(type: SomeType)
task myTask(type: SomeType) { configure closure }
```



Actions

Une tâche est constituée d'une séquence d'objets **Action**.

Lorsque la tâche est exécutée, chaque action est exécutée séquentiellement par l'appel de *Action.execute(T)*.

Des actions peuvent être ajoutées en appelant les méthodes ***doFirst(Action)*** ou ***doLast(Action)***.



Dépendances

Une tâche peut avoir des dépendances sur d'autres tâches.

Gradle garantit le séquençement des tâches dépendantes lors de l'exécution du build

Les dépendances sont contrôlées via les méthodes :


- *dependsOn(Object...), setDependsOn(Iterable)*

L'ordre d'exécution des dépendances par :

- *mustRunAfter(Object...), setMustRunAfter(Iterable),*
- *shouldRunAfter(Object...) setShouldRunAfter(Iterable)*



Task





Propriétés

Chaque instance de *Project* et *Task* fournissent des propriétés accessibles via des méthodes *get/set*

Gradle permet de définir de nouvelles variables sur les classes du modèle via les *extra properties*. (Stockées dans des Map sous la forme clé/valeur)

Pour ajouter des propriétés, l'espace de nom **ext** doit être utilisé. Exemples :

```
project.ext.myProp = 'myValue'
ext {
    someOtherProp = 123
}
assert myProp == 'myValue'
println project.someOtherProp
ext.someOtherProp = 567
```



gradle.properties

Les propriétés peuvent également être directement injectées en les déclarant dans le fichier ***gradle.properties*** placé dans `<USER_HOME>/gradle` ou à la racine du projet.

Elles peuvent être accédées via l'instance du projet.



Propriétés en ligne de commande

Les propriétés peuvent également être fournies :

- via l'option *-P* pour les propriétés projet
- via l'option *-D* pour les propriétés système (Accessible via *systemProp*)
- Ou par des propriétés d'environnement qui suivent le gabarit :
`ORG_GRADLE_PROJECT_propertyName=someValue`



Manipulation de tâches



Déclarer les actions

Une tâche consiste en une séquence d'actions

L'interface *Task* fournit 2 méthodes pour ajouter une action : ***doFirst(Closure)*** et ***doLast(Closure)*** .

```
task printVersion {
    doFirst {
        println "Before reading the project version"
    }
    doLast {
        println "Version: $version"
    }
}
printVersion.doFirst { println "First action" }
printVersion << { println "Last action" }
```



Propriétés par défaut

Une tâche a 2 propriétés par défaut :

- **group** : le groupe auquel la tâche appartient
- **description**

Gradle fournit également une implémentation de logger basé sur SLF4J qui ajoute le niveau QUIET aux niveaux habituels

```
task printVersion {  
    group = 'versioning'  
    description = 'Prints project version.'  
    doLast {  
        logger.quiet "Version: $version"  
    }  
}
```



Définir les dépendances

L'attribut ***dependsOn*** permet de déclarer des dépendances vers une ou plusieurs tâches.

Si plusieurs dépendances sont indiquées,
Gradle ne garantit pas d'ordre d'exécution
entre ces tâches

```
// Pas de garantie sur l'ordre de second et first
task printVersion(dependsOn: [second, first]) << {
    logger.quiet "Version: $version"
}
```



Ordre des tâches

Gradle permet cependant de forcer l'ordre des exécutions des tâches.

Les relations d'ordre sont différentes des relations de dépendances car elle n'influent pas sur ce qu'il doit être exécutée mais seulement l'ordre.

2 méthodes sont disponibles sur Task :

- ***mustRunAfter*** : signifie que la tâche doit s'exécuter après si les 2 tâches sont dans le graphe d'exécution.
- ***shouldRunAfter*** : Moins restrictif. Ignoré si cette règle produit un cycle dans la séquence ainsi qu'en cas de parallélisme



Finalizer

Les tâches ***finalizers*** permettent la libération de ressources après l'exécution d'une autre tâche quelque soit sa réussite.

```
// Second est déclenché par l'exécution de first
task first << { println "first" }
task second << { println "second" }
first.finalizedBy second
```



Ajout de code arbitraire

Du code arbitraire Groovy (script ou classe) peut être ajouté dans le script de build.

On peut par exemple fournir des classes modélisant notre projet et les méthodes associées



Exemple

```
version = new ProjectVersion(0, 1)
class ProjectVersion {
    Integer major
    Integer minor
    Boolean release

    ProjectVersion(Integer major, Integer minor) {
        this.major = major
        this.minor = minor
        this.release = Boolean.FALSE
    }
    ProjectVersion(Integer major, Integer minor, Boolean release) {
        this(major, minor)
        this.release = release
    }

    @Override
    String toString() {
        "$major.$minor${release ? '' : '-SNAPSHOT'}"
    }
}
```



Tâches de configuration

Lors d'ajout de code arbitraire, il est possible de définir des tâches qui ne contiennent pas d'action ... mais des instructions dans leur corps.

Ce sont des tâches de configuration qui sont tout le temps exécutées avant les autres tâches dans la phase de configuration du cycle de build de *Gradle*. (Script Groovy)



Phases Gradle

Lors de l'exécution d'un build, 3 phases distinctes sont exécutées :

- **Initialisation** : Gradle crée une instance de *Project*. Dans un contexte multi-projets, détermine les dépendances entre projets
- **Configuration** : Gradle construit le graphe de tâches prenant part au build. La capacité incrémentale du build détermine les tâches devant être exécutées. Les tâches de configurations sont exécutées
- **Exécution** : Les tâches sont exécutées. L'ordre d'exécution est déterminé par leurs dépendances. Les tâches à jour ne sont pas exécutées.




Exemple tâche de configuration

```
ext.versionFile = file('version.properties')
// Pas d'utilisation de doFirst ou doLast => Tâche de configuration
// Exécutée dans la phase de configuration
task loadVersion {
    project.version = readVersion()
}
ProjectVersion readVersion() {
    logger.quiet 'Reading the version file.'
    if(!versionFile.exists()) {
        throw new GradleException("Required version file does not exist:
$versionFile.canonicalPath")
    }

    Properties versionProps = new Properties()
    versionFile.withInputStream { stream -> versionProps.load(stream) }

    new ProjectVersion(versionProps.major.toInteger(), versionProps.minor.toInteger(),
        versionProps.release.toBoolean())
}
```

Cycle de vie





Build incrémental

Gradle détermine si une tâche est à jour en comparant les entrées et les sorties d'une tâche entre 2 builds


Les entrées/sorties (*input/output*) sont des champs de la classe *Task*

- Une entrée peut être un répertoire, un ou plusieurs fichiers ou une propriété arbitraire
- Une sortie peut être un répertoire ou un ensemble de fichiers

L'initialisation des entrées/sorties est effectuée pendant la phase de configuration et doit donc être placée dans le corps de la tâche



TaskInputs et TaskOutputs





Exemple

```
task makeReleaseVersion(group: 'versioning', description: 'Makes project
    a release version.') {
    // Configuration phase
    // l'entrée est une propriété, la sortie le fichier modifié par la tâche
    inputs.property('release', version.release)
    outputs.file versionFile

    doLast {
        version.release = true
        ant.propertyfile(file: versionFile) {
            entry(key: 'release', type: 'string', operation: '=', value:
            'true')
        }
    }
}
```




Custom Task

En utilisant le DSL de *Gradle*, les tâches instanciées sont de type ***DefaultTask***

Si l'on veut contrôler complètement l'exécution d'une tâche, on peut créer sa propre implémentation



Exemple

```
class ReleaseVersionTask extends DefaultTask {  
    // Déclaration des entrées/sorties par annotation  
    // Entrée requise sinon : TaskValidationException, (Voir @Optional)  
    @Input Boolean release  
    @OutputFile File destFile  
  
    ReleaseVersionTask() {  
        group = 'versioning'  
        description = 'Makes project a release version.'  
    }  
    @TaskAction  
    void start() {  
        project.version.release = true  
        ant.propertyfile(file: destFile) {  
            entry(key: 'release', type: 'string', operation: '=', value: 'true')  
        }  
    }  
}
```



Utilisation

Pour utiliser la classe, il faut définir une tâche du type désiré et initialiser ses propriétés dans le bloc de configuration.

```
task makeReleaseVersion(type: ReleaseVersionTask) {  
    release = version.release  
    destFile = versionFile  
}
```



Tâche prédéfinies Gradle

Gradle fournit beaucoup d'implémentations de l'interface *Task*

- Copy, Zip, Jar, Exec, tar, Delete
- JavaCompile, Javadoc, JavaExec
- Jar, war, ear
- Jacoco, Checkstyle, FindBugs
- ...

Voir : <https://docs.gradle.org/current/dsl/index.html>



Exemple

```
task makePretty(type: Delete) {  
    // Appel de la méthode  
    // pendant la phase de configuration  
    delete 'uglyFolder', 'uglyFile'  
}
```

La tâche *Delete* contient une propriétés *delete* qui liste les fichiers,répertoires à supprimer

Elle expose la méthode *delete(args)* qui ajoute les arguments à la propriétés *delete*



Exemple

```
task createDistribution(type: Zip, dependsOn: makeReleaseVersion) {  
    // référence implicite à la sortie de la tâche War  
    // => Une dépendance est déduite par Gradle  
    from war.outputs.files  
    // Les fichiers sources sont placés dans le répertoire src du fichier Zip  
    from(sourceSets*.allSource) {  
        into 'src'  
    }  
    // Ajouter le fichier de version au zip  
    from(rootDir) {  
        include versionFile.name  
    }  
}  
task backupReleaseDistribution(type: Copy) {  
    from createDistribution.outputs.files  
    into "$buildDir/backup"  
}  
task release(dependsOn: backupReleaseDistribution) << {  
    logger.quiet 'Releasing the project...'  
}
```



Dépendances multiples

Les relations de dépendances entre tâches construisent le graphe acyclique de Gradle

Lorsque une tâche dépend de plusieurs tâche, l'attribut *dependsOn* est initialisé avec un tableau ... mais il n'y a pas de garantie d'ordre sur les tâches dont on dépend

```
// Pas de garantie d'ordre pour l'exécution des tâches V et W
task taskX (group: 'myGroup', description: 'The taskX.',
    dependsOn: [taskV, taskW] ) {
    doLast { println 'taskX' }
}
```



Task rule

Gradle définit **task rule** qui exécute de la logique basée sur un gabarit de nom de tâche composé :

- D'une partie statique
- D'un « placeHolder » permettant d'influencer la logique

Exemple :

Gabarit : *increment<Classifier>Version*

Usage : *incrementMajorVersion*


La définition de task rule s'effectue sur une classe *TaskContainer* en utilisant la variable **tasks**



Example

```
tasks.addRule("Pattern: ping<ID>") { String taskName ->
    if (taskName.startsWith("ping")) {
        task(taskName) {
            doLast {
                println "Pinging: " + (taskName - 'ping')
            }
        }
    }
}
```

```
---
> gradle -q pingServer1
Pinging: Server1
```



Sources pour le build

Le répertoire ***buildSrc*** peut être une alternative pour placer du code de build.

Le code peut alors être organisé en package et peut contenir des classes de test.

Gradle propose une arborescence pour les sources de build. Tout code trouvé dans ces répertoires est ajouté au classpath du script de build

- *src/main/java*
- *src/main/groovy*

Il faut alors importer les packages dans le script de build



Hooks



Callback

Il y a 2 façons d'exécuter du code de réagissant à des événements de build (*callback*):

- Avec une closure
- En implémentant une interface *listener* fournie par l'API *Gradle* .

De nombreux hooks sont fournis par Gradle (Voir l'api des classes *Project* et *Gradle*)



org.gradle.api.invocation.Gradle

addBuildListener(buildListener) : Ajout d'une implémentation de *BuildListener*. Événements lors de l'exécution du build.

addProjectEvaluationListener(listener)
Listener lors de l'évaluation des projets


afterProject(closure) : Ajout d'une closure après l'évaluation d'un projet



Gradle

L'objet *Gradle* permet également d'accéder au graphe dirigé des tâches (avec toutes les dépendances calculés)

Il est possible alors d'utiliser la méthode *whenReady* pour ajouter un traitement par Closure





Example

```
task distribution {
    doLast {
        println "We build the zip with version=$version"
    }
}

task release(dependsOn: 'distribution') {
    doLast {
        println 'We release now'
    }
}

gradle.taskGraph.whenReady {taskGraph ->
    if (taskGraph.hasTask(release)) {
        version = '1.0'
    } else {
        version = '1.0-SNAPSHOT'
    }
}
```



Plugins



Introduction

Toutes les fonctionnalités utiles de *Gradle* (comme compiler du code Java) sont apportées par les plugins.

Les plugins

- Ajoute de nouvelles tâches (Ex : *JavaCompile*)
- De nouveaux objets du modèle (Ex : *SourceSet*)
- Des conventions (Ex : Code source dans *src/main/java*)
- Des extensions aux objets cœur ou provenant d'autres plugins .



Types de plugins

2 types de plugins sont possibles dans Gradle

- Les **plugins script** sont des scripts de build additionnels qui ajoutent des détails de configuration et utilisent une approche déclarative pour modifier un build final. Ils sont typiquement utilisé à l'intérieur d'un script de build
- Les **plugins binaires** sont des classes qui implémentent l'interface *Plugin* et adoptent une approche programmatique pour manipuler le build.
Typiquement fourni via un jar.



Utilisation d'un plugin

Lors de l'utilisation d'un plugin, Gradle doit :

- Le localiser
- L'appliquer. Exécuter *Plugin.apply(T)*

Cela est fait en une seule fois via une syntaxe déclarative légèrement différente en fonction du type du plugin



Plugin script

Les plugins script sont automatiquement résolus et peuvent être appliqués à partir d'un script présent

- en local : Emplacement relatif au répertoire projet
- ou en distant : Spécifié avec une URL HTTP

```
apply from: 'other.gradle'
```




Plugin binaire

Les plugins binaires sont appliqués grâce à leur ***plugin id***.

Les plugins cœur de Gradle fournissent des noms courts (comme 'java').

Les autres fournissent un nom qualifié (comme *com.github.foo.bar*)



Emplacements

Les plugins binaires doivent être résolus avant d'être appliqués. Cela peut être fait de différentes façons :

- Inclure le plugin à partir du portail Gradle ou d'un dépôt personnalisé en utilisant le DSL ***plugins***
- Inclure le plugin à partir d'un jar externe défini comme dépendance externe du script de build
- Définir le plugin comme un fichier source du répertoire *buildSrc* du projet
- Définir le plugin comme une classe interne au script de build



DSL *plugins*

L'utilisation du DSL *plugins* apporte certains avantages :

- Optimise le chargement et à la réutilisation des classes du plugin
- Permet à différents plugins d'utiliser différentes versions des dépendances
- Permet l'assistance lors de l'édition

La syntaxe est :

```
plugins {  
    id «plugin id» version «plugin version» [apply «false»]  
}
```

Aucun autre code n'est permis dans le bloc plugins

La résolution se fait du vis à vis du portail Gradle. Exemple :

```
plugins {  
    id 'java'  
    id 'com.jfrog.bintray' version '0.4.1'  
}
```



Plugin Management

Il est possible de spécifier des dépôts de plugins spécifiques en utilisant le bloc ***repositories {}*** à l'intérieur d'un bloc ***pluginManagement {}*** dans le fichier *settings.gradle*

```
pluginManagement {  
    repositories {  
        maven {  
            url 'maven-repo'  
        }  
        gradlePluginPortal()  
        ivy {  
            url 'ivy-repo'  
        }  
    }  
}
```




Legacy plugin

Si ce n'est pas possible d'utiliser DSL, on peut encore utiliser l'ancienne méthode qui n'est pas pré-compilé

Appliquer via un id

```
apply plugin: 'java'
```

Ou via la classe

```
apply plugin: JavaPlugin
```



Jar externe

Les plugins binaires qui ont été packagés sous forme de *jar* peuvent être ajouté au *classpath* du script de build puis appliqué. Il faut utiliser alors la méthode ***buildScript()*** Exemple :

```
buildscript {  
    repositories {  
        jcenter()  
    }  
    dependencies {  
        classpath "com.jfrog.bintray.gradle:gradle-bintray-plugin:0.4.1"  
    }  
}  
  
apply plugin: "com.jfrog.bintray"
```



Écriture de plugin

```
class GreetingPlugin implements Plugin<Project> {  
    void apply(Project project) {  
        project.task('hello') {  
            doLast {  
                println 'Hello from the GreetingPlugin'  
            }  
        }  
    }  
}
```

```
// Apply the plugin  
apply plugin: GreetingPlugin
```



Plugin *init*

Le plugin *init* automatiquement disponible peut être utilisé pour créer un nouveau build gradle.

Il permet la création ou la conversion de différents types de projet

Il permet également la génération de wrapper permettant une installation automatique de gradle sur d'autres postes

gradle init



Wrapper

Le **wrapper** permet d'exécuter un script Gradle sans avoir à installer le runtime

- Pour sa mise en place, il suffit d'exécuter dans le répertoire projet :
gradle wrapper

Il garantit également que le script de build est exécuté avec une version spécifique de Gradle.

- Il télécharge automatiquement le runtime Gradle à son premier lancement et le stocke localement
\$HOME_DIR/.gradle/wrapper/dists

=> L'objectif étant de créer des build reproductibles et surs quelque soit l'OS, ou la version de *Gradle* installée manuellement



Types d'initialisation

init supporte différents type de mise en place via l'argument *--type*

- pom (Maven conversion)
- java-application, java-library
- Scala-library
- groovy-library, groovy-application
- basic



Java et C++

Gestion des dépendances
Plugin Java
Plugin C++



Gestion des dépendances





Introduction

L'ajout de dépendances vers des librairies externes s'effectue en 2 étapes :

- Avec la closure ***dependencies***, on définit les libraires dont on dépend ainsi que leur *scope* ou configuration
- Avec la closure ***repositories***, on indique l'origine de ces dépendances .

Gradle résout alors automatiquement les dépendances, les télécharge et les stocke dans son cache local





Configuration

Les **configurations** définissent le périmètre d'une dépendance.


Chaque projet contient une classe *ConfigurationContainer* qui gère les configurations d'un projet.

Une configuration regroupe les dépendances pour une fonctionnalité spécifique.

Par exemple, le plugin Java fournit 6 configurations : *compile* , *runtime* , *testCompile* , *testRuntime* , *archives* et *default*



Interface Configuration





Définir une configuration

On peut définir ses propres configurations via le DSL ***configurations***.

Une configuration consiste d'un nom et d'une visibilité (cas multi-projets)

La configuration peut ensuite être accédée via son nom à partir de la variable ***configurations*** pour référencer un classpath par exemple

.




Exemple

```
configurations {
    jasper
}

repositories {
    mavenCentral()
}

dependencies {
    jasper 'org.apache.tomcat.embed:tomcat-embed-jasper:9.0.2'
}

task preCompileJsps {
    doLast {
        ant.taskdef(classname: 'org.apache.jasper.JspC',
                    name: 'jasper',
                    classpath: configurations.jasper.asPath)
        ant.jasper(validateXml: false,
                    uriroot: file('src/main/webapp'),
                    outputDir: file("$buildDir/compiled-jsps"))
    }
}
```



Héritage des configurations

Une configuration peut étendre une autre configuration.

Les configurations filles héritent de toutes les dépendances de ses parents.

```
configurations {  
    smokeTest.extendsFrom testImplementation  
}  
  
dependencies {  
    testImplementation 'junit:junit:4.12'  
    smokeTest 'org.apache.httpcomponents:httpclient:4.5.5'  
}
```



Bloc *dependencies*


Le bloc ***dependencies*** est utilisé pour affecter une ou plusieurs dépendances à une configuration

Les dépendances peuvent être exprimées vis à vis :

- D'un artefact d'un dépôt
- D'un projet
- D'un ensemble de fichiers du système de fichiers



Interface *Dependency*





Coordonnées d'une dépendance

Lorsque le gestionnaire de dépendances recherche une dépendance dans un dépôt, il le localise avec les attributs suivants :

- **group** : Habituellement une organisation. Peut utiliser un point dans son nom.
- **name** : Le nom de l'artefact.
- **version** : Une chaîne de caractère composée d'une version majeure et mineure
- **classifier** : Éventuellement, pour distinguer les artefacts avec les mêmes groupes, nom et version

La syntaxe est alors :

```
dependencies {  
    configurationName dependencyNotation1, dependencyNotation2, ...  
}
```



Rapport de dépendances

Lorsque l'on exécute :
gradle dependencies

L'arbre complet de dépendances est affiché

On y voit pour chaque configuration les dépendances directes et les dépendances transitives

Les dépendances omises sont marquées d'un astérisque.

- Cela est dû à une double dépendance vers le même artefact.
- Par défaut, *Gradle* choisit la version la plus haute en cas de disparité dans les versions



Exclure des dépendances transitives

La méthode ***exclude*** avec comme paramètre une *Map* permet d'exclure des dépendances transitives

Exemple :

```
dependencies {  
    cargo('org.codehaus.cargo:cargo-ant:1.3.1') {  
        exclude group: 'xml-apis', module: 'xml-apis'  
    }  
    cargo 'xml-apis:xml-apis:2.0.2'  
}
```

L'attribut ***transitive*** permet d'exclure toutes les dépendances transitives

```
dependencies {  
    cargo('org.codehaus.cargo:cargo-ant:1.3.1') {  
        transitive = false  
    }  
    // Déclarer explicitement toutes les dépendances nécessaires  
}
```



Version dynamique

La déclaration de version dynamique a plusieurs syntaxes :

// Dernière version

```
org.codehaus.cargo:cargo-ant:latest-integration
```

// Version 1.x

```
org.codehaus.cargo:cargo-ant:1.+
```

// Version comprise entre 1.0 et 2.0 inclus

```
org.codehaus.cargo:cargo-ant:[1.0,2.0]
```


// Version comprise entre 1.0 inclus et 2.0 exclus

```
org.codehaus.cargo:cargo-ant:[1.0,2.0[
```

// Version à partir de 1.0

```
org.codehaus.cargo:cargo-ant:[1.0, )
```

Interface *Repository*





Dépôts

Gradle supporte les dépôt les plus habituels (maven, google, ...)

En dehors des dépôts centraux pré-configurés, il est possible de fournir une URL arbitraire, une référence vers le système de fichier local, des informations d'authentification

Dans le bloc *repositories*, on peut déclarer plusieurs dépôts, ils seront interrogés par Maven dans leur ordre de déclaration.




Dépôts publics classiques

L'interface *RepositoryHandler* fournit plusieurs méthodes permettant de définir les dépôts habituels :

- Maven
 - `mavenCentral()`
 - `mavenLocal()`
- Bintray JCenter
 - `jcenter()`
- Dépôt de Google Android
 - `google()`

Raccourci





Dépôt Maven custom

L'API Gradle permet de configurer un repository Maven via la méthode ***maven()***

Exemple

```
repositories {  
    maven {  
        credentials {  
            username = 'joe'  
            password = 'secret'  
        }  
        url  
        "http://localhost:8081/nexus/content/groups/public"  
    }  
}
```



Cache Gradle

Le répertoire racine pour stocker les artefacts en local est :

`<USER_HOME>/ .gradle/caches`

2 types de stockage sont gérés :

- Basé sur les fichiers (jars ou méta-données comme *pom.xml*). Le chemin de stockage contient un checksum SHA1 .
- Un stockage binaire contenant le résultat de la résolution de versions dynamiques et autres.



Plugin Java



Introduction

Le plugin **Java** ajoute les tâches de compilation et de tests d'un projet.

Il sert de base pour de nombreux autres plugins Gradle

```
apply plugin: 'java'
```



Structure projet

src/main/java : Source de production Java


src/main/resources : Ressources de production

src/test/java : Classes de test

src/test/resources : Ressources de test

src/<sourceSet>/java : Sources destinés à autres chose que la production ou les tests unitaires

src/<sourceSet>/resources : Ressources destinées à autres chose que la production ou les tests unitaires



SourceSet

Le plugin ajoute donc 2 ***sourceSet*** :

- ***main*** : Le code de production assemblé dans un JAR
- ***test*** : Le code de test compilé et exécuté avec JUnit ou TestNG. Typiquement des tests unitaires ou tout autre tests qui utilise le même classpath.

Il est possible de :

- changer le layout des 2 sourceSet prédéfinis
- de rajouter d'autres *sourceSet*. Comme *integration* par exemple pour les tests d'intégration



Exemple : Chgt de layout

```
sourceSets {  
    main {  
        java {  
            srcDirs = ['src/java']  
        }  
        resources {  
            srcDirs = ['src/resources']  
        }  
    }  
}
```




Exemple : ajout de *sourceSet*

```
sourceSets {  
    intTest {  
        java.srcDir file('src/intTest/java')  
        resources.srcDir file('src/intTest/resources')  
        compileClasspath += sourceSets.main.output +  
configurations.testRuntime  
        runtimeClasspath += output + compileClasspath  
    }  
}  
  
configurations {  
    intTestImplementation.extendsFrom implementation  
}  
  
dependencies {  
    intTestImplementation 'junit:junit:4.12'  
}
```



Interface *SourceSet*

name : Le nom

java.srcDirs : Le répertoire source.

resources.srcDirs : Le répertoire ressource

output.classesDirs : Le répertoire pour générer les classes

output.resourcesDir : Le répertoire pour généré les
ressources

compileClasspath : Le classpath de compilation.

runtimeClasspath : Le classpath d'exécution .

allJava : Tous les fichiers *.java*

allSource : Toutes les sources



Tâches

compileJava(type: JavaCompile) Compile les fichiers sources de production avec *javac*. Dépend de toutes les tâches produisant le classpath (la tâche *jar* des dépendances).

processResources(type: Copy) Copies les ressources de production dans le répertoire *resources*.

classes(type: Task) Assemble les répertoire des classes de production et de ressources.

compileTestJava(type: JavaCompile) Compiles les classes de test avec *javac*. Depend de *compile*, plus toutes les tâches produisant le classpath de *testCompile*.

processTestResources(type: Copy) Copie les ressources de test dans le répertoire *test resources directory*.

testClasses(type: Task) : Assembles les répertoires des classes et ressources de test. Dépend de *compileTestJava* et *processTestResources*.



Tâches (2)

jar(type: Jar) : Assembles le fichier JAR. Dépend de *compile*.

javadoc(type: Javadoc) : Génère la documentation avec Javadoc. Dépend de *compile*.

test(type: Test) : Exécute les tests unitaires avec JUnit ou TestNG. Dépend de *compile*, *compileTest*, plus toutes les tâches qui produisent le classpath test runtime.

uploadArchives(type: Upload) : Upload les artefacts dans la configuration archives. Dépend des tâches qui produisent les artefacts.

clean(type: Delete) : Supprime le répertoire de build .

cleanTaskName(type: Delete) : Supprime les sorties d'une tâche



Tâches du cycle de vie

assemble(type: Task) : Assemble toutes les archives d'un projet.

check(type: Task) : Effectue toutes les tâches de vérification d'un projet. Dépend en particulier des tests.

build(type: Task) : Build du projet. Dépend de *check* et *assemble*.

buildNeeded(type: Task) : Effectue un build complet du projet et de tous les projets dont il dépend. Dépend de *build* et *buildNeeded* dans les dépendances projet.

buildDependents(type: Task) : Effectue un build complet du projet et de tous les projets qui en dépendent.

buildConfigName(type: Task) : Assemble les artefacts de la configuration spécifiée.

uploadConfigName(type: Upload) : Upload les artefacts de la configuration spécifiée



Tâches liées à un sourceSet

compileSourceSetJava(type: JavaCompile) :


Compile le *sourceSet* spécifié.

processSourceSetResources(type: Copy) :

Copie les ressources du *sourceSet* spécifié dans le répertoire des ressources.

sourceSetClasses(type: Task) : Assemble les répertoire des classes de production et de ressources du *sourceSet* spécifié.

Dépendances





Configurations de dépendances

Le plugin Java ajoute des configurations de dépendances

- ***implementation, compile(Déprécié)*** : Les dépendances de compilation,
- ***compileOnly*** : Les dépendances présentes seulement à la compilation
- ***compileClasspath*** *extends compile, compileOnly, implementation* : Classpath utilisé à la compilation
- ***annotationProcessor*** : Processeurs d'annotations utilisés pendant la compilation
- ***testImplementation, testCompile(Déprécié)*** *extends compile* : Dépendances additionnelles pour compiler les tests.
- ***testCompileOnly*** : Dépendances additionnelles pour compiler les tests mais pas les exécuter
- ***testCompileClasspath*** *extends testCompile, testCompileOnly, :* Classpath utilisé par la tâche *compileTestJava*.



Configurations de dépendances (2)

runtimeOnly, runtime(déprécié) extends compile : Les dépendances à l'exécution

runtimeClasspath extends *runtimeOnly, runtime, implementation* :
Implementation + runtime.

testRuntimeOnly, testRuntime(Deprecated) extends *runtime, testCompile* : Dépendances additionnelles pour exécuter les tests.

testRuntimeClasspath extends *testRuntimeOnly, testRuntime, testImplementation* : Classpath d'exécution pour l'exécution des tests.

archives : Artefacts produit par ce projet. Utilisé par les tâches *uploadArchives*.

default extends *runtime* : La configuration par défaut, contient les artefacts et les dépendances nécessaires à l'exécution.



Configuration pour un *sourceSet*

sourceSetImplementation, sourceSetCompile(Deprecated) :

Dépendances de compilation pour le sourceSet spécifié

sourceSetCompileOnly : Compilation mais pas exécution

sourceSetCompileClasspath *extends compileSourceSetJava*

sourceSetAnnotationProcessor : Processeur d'annotation
utilisé durant la compilation

sourceSetRuntimeOnly, sourceSetRuntime(Deprecated) :
Dépendances exclusivement runtime pour le source set spécifié.

sourceSetRuntimeClasspath *extends sourceSetRuntimeOnly, sourceSetRuntime, sourceSetImplementation* : Classpath résolu
lors de l'exécution pour ce sourceSet



Répertoires

reporting.baseDir : Le nom répertoire pour générer des rapport
(*reports*)

reportsDir : Le répertoire résolu ou sont générer les rapports
(*buildDir/reports*)

testResultsDir Le répertoire résolu ou sont générer les résultats de test
en .xml (*buildDir/test-results*)

testReportDir : Le répertoire pour générer le rapport de test.
(*reportsDir/tests*)

libsDir : Le répertoire pour générer les librairies (*buildDir/libs*)

distsDir Le répertoire pour générer la distribution. (*buildDir/distributions*)

docsDir Le répertoire pour générer la documentation (*buildDir/docs*)

dependencyCacheDirName : le nom du répertoire pour cacher les
informations de dépendances des sources (*dependency-cache*)



Autres propriétés

sourceSets : Tous les sourceSets

sourceCompatibility: Version pour le compilateur. Par défaut la version de la JVM courante. Supporte des String ou Nombres

targetCompatibility : Version pour la génération de classe. Par défaut *sourceCompatibility*

archivesBaseName : Le nom de base pour générer les archives (JAR ou ZIP). Par défaut le nom du projet.

manifest : Le manifeste à inclure dans le JAR. Par défaut vide.



Plugin Java library

Le plugin *java-library* étend le plugin Java

Une librairie est un composant Java consommé par d'autres composants.

Le plugin offre alors 2 nouvelles configurations de dépendances :

- ***api*** : Les classes publiques
- ***implementation*** : Les classes internes



Plugin *application*

Le plugin ***application*** permet de désigner la classe contenant la méthode main pouvant être exécutée en ligne de commande.

```
apply plugin: 'java'  
apply plugin: 'application'
```

```
mainClassName = 'App'
```



Application web

Gradle fournit le plugin **war** ; un plugin communautaire **gretty** permet de tester et déployer une application sur *Jetty* ou *Tomcat*

Le plugin *War* étend le plugin *Java* en remplaçant la génération du JAR par la génération d'un WAR.

La structure par défaut du projet contient un répertoire **src/main/webapp** contenant les ressources web du projet

```
plugins {  
    id 'war'  
    id 'org.akhikh1.gretty' version '1.4.2'  
}
```



Plugin C++



Introduction

Gradle supporte les outils *Clang*, *GCC* et *Visual C++*

Il propose plusieurs plugins :

- ***cpp-application*** : Pour les applications
- ***cpp-library*** : Pour les bibliothèques
- ***cpp-unit-test*** : Pour les tests unitaires




Tâches des plugins

Les plugins ***cpp-application*** et ***cpp-library*** ajoutent les tâches de compilation, de link et d'assemblage :


- ***compileDebugCpp*** et ***compileReleaseCpp*** qui compilent les fichiers source sous *src/main/cpp*.
- ***linkDebug*** et ***linkRelease*** qui lie les fichiers d'objets compilés en un exécutable.
- ***createDebug*** et ***createRelease*** qui assemblent les fichiers objets C ++ compilés dans une bibliothèque statique

Ces tâches sont intégrées aux tâches du cycle de vie standard.

Par exemple, la tâche produisant le binaire final est associée à ***assemble***. Par défaut, c'est la version de debug




Graphe de dépendances des tâches






Graphe de dépendances pour bibliothèques

Librairie partagée



Librairie statique





Différence entre library et application

Les projets **library** sont consommées par d'autres projets C++

Les méta-données de dépendances sont publiées avec les binaires et les entêtes

Elles distinguent les dépendances :

- qui sont uniquement nécessaires pour compiler la bibliothèque
- De celles qui sont également requises pour compiler le consommateur : configuration **api**

Les projets **application** sont destinés à produire des exécutables

Le plugin permet :

- Une tâche **install** qui crée un répertoire permettant d'exécuter l'application
- Des scripts permettant de démarrer l'application



Variantes de build

Généralement, les projets natifs produisent différents binaires (plateforme, architecture processeur, debug, ...)

Gradle gère cela à travers les concepts de **dimensions** et de **variantes**.


- Une dimension est simplement une catégorie.
Par exemple, pour les applications C++
"build type" : *debug* ou *release*
"Machine Cible" : linux.x86_64, windows.x86
- Une variante est une combinaison de valeurs de dimension.
Par exemple : windows.x86/Debug



Machines cibles

Par défaut, *Gradle* crée un binaire pour l'OS hôte et son architecture. Il est possible de remplacer cela en spécifiant un ensemble de machines cible dans les blocs *application* ou *library*:

```
application {  
    targetMachines = [  
        machines.linux.x86_64,  
        machines.windows.x86, machines.windows.x86_64,  
        machines.macOS.x86_64  
    ]  
}
```



Configuration des sources

Par défaut, les sources sont présents dans :

- ***src/main/cpp***
- Les entêtes privées dans ***src/main/headers***
- Pour les librairies, les entêtes publiques dans ***src/main/public***

Les emplacement peuvent être modifiés dans les blocs
application ou *library*

Exemple :

```
library {  
    source.from file('src')  
    privateHeaders.from file('src')  
    publicHeaders.from file('include')  
}
```




Dépendances

Gradle permet de définir des dépendances vers des référentiels Maven ou vers un autre projet dans les blocs *application* ou *library*

```
application {  
    dependencies {  
        implementation project(':common')  
        implementation 'org.gradle.cpp-samples:list:1.5'  
    }  
}
```



Configurations

Les principales configurations pour les dépendances sont :

- ***implementation*** : Compilation, linking et exécution
- ***cppCompileVariant*** : Dépendances nécessaires pour la compilation mais pas pour le link et l'exécution
- ***nativeLinkVariant*** : Seulement pour le link
- ***nativeRuntimeVariant*** : Seulement pour l'exécution
- ***api*** : (Apporté par le plugin *cpp-library*) Dépendances nécessaires pour la compilation et le link du module courant et des modules qui dépendent du module courant.



Autres cas d'usage

Définition d'un référentiel personnalisé compatible Maven

Déclarer des dépendances avec des versions changeantes (par exemple SNAPSHOT) et dynamiques (plage)

Contrôle des dépendances transitives et de leurs versions

Test de vos correctifs pour la dépendance tierce via des [build composites](#).



Compilation et link

La compilation et le link d'un projet C++ est facile si :

- On place le code source dans ***src/main/cpp***
- On déclare les dépendances dans le scope ***implementation***
- On exécute la tâche ***assemble***



Options de compilation et de link

Les options de compilation et du link accessibles via les tâches *compileVariantCpp*, *linkVariant* et *createVariant* qui sont de type *CppCompile*, *LinkSharedLibrary* et *CreateStaticLibrary* respectivement.

```
tasks.withType(CppCompile).configureEach {  
  
    // Une option de compilation  
    compilerArgs.add '-W3'  
  
    // Des options spécifiques à certains outils  
    compilerArgs.addAll toolChain.map { toolChain ->  
        if (toolChain in [ Gcc, Clang ]) {  
            return ['-O2', '-fno-access-control']  
        } else if (toolChain in VisualCpp) {  
            return ['/Zi']  
        }  
        return []  
    }  
}
```



Accès aux variantes


Il est également possible de trouver l'instance pour une variante spécifique via ***binaries*** de type BinaryCollection dans un bloc *application* ou *library* :

```
application {  
    binaries.configureEach(CppStaticLibrary) {  
        // Définir une option de compilation  
        compileTask.get().compilerArgs.add '-W3'  
  
        // En fonction de la toolChain  
        if (toolChain in [ Gcc, Clang ]) {  
            compileTask.get().compilerArgs.addAll(['-O2', '-fno-access-control'])  
        } else if (toolChain in VisualCpp) {  
            compileTask.get().compilerArgs.add('/Zi')  
        }  
    }  
}
```



Test

Le plugin ***cpp-unit-test*** fournit les tâches, les configurations et les conventions pour l'intégration avec un framework de test tel que Google Test.






Bloc unit-test

Le plugin ajoute le bloc ***unitTest*** dans lequel on peut définir :

- Les machines cibles
- Les dépendances
- ...

Exemple :

```
unitTest {  
    targetMachines = [  
        machines.linux.x86_64,  
        machines.windows.x86, machines.windows.x86_64,  
        machines.macOS.x86_64  
    ]  
}
```

Packaging et publication

Par défaut,

- Les fichiers exécutables sont publiés directement dans des référentiels Maven.
- Les librairies partagés et statiques sont publiés dans les référentiels Maven avec un zip des en-têtes publics.
- Pour les applications, Gradle permet l'installation et l'exécution de l'exécutable avec toutes ses dépendances de bibliothèque partagée dans un emplacement spécifié.



Multi-projets



Structure

Un multi-projets consiste de plusieurs répertoires dont un répertoire contient le projet parent.

2 dispositions sont possibles :

- Arborescente : les sous-projets sont des sous-répertoires du projet parent
- A plat : le projet parent et les sous-projets sont au même niveau, le projet parent s'appelle *master*

Le projet parent coordonne la construction des sous-projets et peut définir des comportements communs ou spécifiques aux sous-projets

Il contient également l'unique répertoire de code de build *buildSrc*




Phase de configuration

La phase de configuration d'un projet consiste à exécuter le script *build.gradle*

- Par défaut, la configuration de tous les sous-projets survient avant l'exécution de toute tâche (même si on ne désire qu'exécuter une seule tâche d'un sous-projet). Pour de gros projets, ceci peut être pénalisant.

A partir de Gradle 1.4, le mode **configuration à la demande** a été introduit. Il a pour but de ne configurer que les projets nécessaires à l'exécution d'une tâche particulière.

- Propriété de *gradle.properties* :
`org.gradle.configureondemand=true`
- A terme seul ce mode sera effectif par défaut



Déclaration des sous-projets


Le fichier ***settings.gradle*** dans le répertoire du projet parent liste les sous-projets via la méthode ***include***

```
include 'model'  
include 'repository', 'web'
```

La ligne de commande *gradle projects* liste l'arborescence projet

settings.gradle est accessible
programmatically via la variable ***settings***

Interface *Settings*





Résolution des *Settings*

Gradle permet d'exécuter le build à partir du projet racine ou des répertoires des sous-projets.

Gradle recherche le fichier *settings* en 2 étapes :

- 1ère étape il recherche un répertoire nommé *master* au même niveau que le répertoire d'exécution
- Sinon, il remonte dans l'arborescence



Comportement commun

L'API *Project* offre les méthodes ***allprojects*** et ***subprojects*** permettant de définir les comportements communs aux sous-projets.

Exemple :

```
allprojects {  
    task hello {  
        doLast { task -> println "I'm $task.project.name" }  
    }  
}  
subprojects {  
    hello {  
        doLast { println "- I depend on water" }  
    }  
}
```




Comportement spécifique

Les comportements spécifiques peuvent être spécifiés dans le fichier *build.gradle* des sous-projets mais également dans le projet racine avec la méthode ***project***

```
project(':bluewhale').hello {  
    doLast {  
        println "- I'm the largest animal on this planet."  
    }  
}
```



Autre exemple

build.gradle

```
allprojects {  
    task hello {  
        doLast { task -> println "I'm $task.project.name" }  
    }  
}
```

bluewhale/build.gradle

```
hello.doLast { println "- I'm the largest animal on this planet." }
```

gradle -q hello

```
> gradle -q hello  
I'm water  
I'm bluewhale  
- I depend on water  
- I'm the largest animal on this planet.
```



Méthode *configure()*

La méthode ***configure()*** prend comme argument une liste de projets et applique la configuration à tous les projets. Exemple :

```
configure(subprojects.findAll {it.name != 'tropicalFish'}) {  
    hello {  
        doLast {  
            println '- I love to spend time in the arctic  
waters.'  
        }  
    }  
}
```



Configuration selon des propriétés

En utilisant les méthodes de callback d'un build, on peut configurer des projets après l'évaluation des propriétés.

Exemple :

```
subprojects {
    hello {
        doLast {println "- I depend on water"}
        afterEvaluate { Project project ->
            if (project.arctic) { doLast {
                println '- I love to spend time in the arctic waters.' }
            }
        }
    }
}
```



Règle d'exécution des tâches

Lors de l'indication d'une tâche, la ligne de commande *gradle* parcourt l'arborescence en partant du répertoire courant et exécute toutes les tâches du nom spécifié dans les différents sous-projets.

Pour exécuter une seule tâche d'un sous-projet, il faut indiquer le chemin absolu

Exemple :

```
$ gradle :model:build
```



Dépendances entre sous-projets

Plusieurs types de dépendances existent entre les sous-projets

- Dépendance sur la configuration. Ex : Le parent qui injecte sa configuration chez les enfants
- Dépendance à l'exécution, influant sur l'ordre d'exécution des tâches
 - Dépendance implicite à cause des entrées/sorties des tâches
 - Dépendance explicite en utilisant *dependsOn*
 - Dépendance explicite, un sous-projet a besoin d'un autre pour un *classpath*

Si il n'y a aucune dépendance, Gradle exécute les tâches dans l'ordre alphabétique



Dépendances vers un classpath d'un sous-projets

Déclarer une dépendance vers un sous-projet est similaire à la déclaration d'une dépendance vers une librairie. Il faut utiliser *dependencies* et préciser la configuration. Exemple :

settings.gradle

```
include 'web', 'repository', 'model'
```

build.gradle

```
project(':repository') {  
    dependencies {  
        implementation project(':model')  
    }  
}  
  
project(':web') {  
    dependencies {  
        implementation project(':repository'), project(':model')  
    }  
}
```




Dépendances et exécution des tâches

Lors de l'exécution d'une tâche particulière, toutes les tâches du même nom des dépendances d'un sous-projet sont exécutées en premier.

Exécuter des tâches du projet racine, permet de s'assurer que toutes les classes dont les projets sont à jour.

\$ gradle build





Build partiel

Pour des multi-projets volumineux, *Gradle* fournit une fonctionnalité nommée build partiel via l'option **-a** ou **-no-rebuild**

Cela permet à *Gradle* de ne pas vérifier si les projets dont on dépend doivent être reconstruits

Par exemple, si on sait que les classes de *model* n'ont pas changé

```
$ gradle :repository:build -a
```



Forcer l'exécution des tests des dépendances

Autre cas, si on travaille seulement sur le projet *repository* et que l'on effectue `gradle:repository:build`

- Cela a pour effet de provoquer la compilation du projet dépendant *model* mais pas l'exécution des tests.

Pour cela :

`gradle :repository:buildNeeded`



buildDependents

Avec la tâche ***buildDependents***, on peut vérifier l'impact de ses modifications en construisant et testant les projets dépendants.

Exemple :

```
gradle:repository:buildDependents
```



Dépendances entre tâches des sous-projets

Il est possible de déclarer une dépendance vers une tâche d'un autre sous-projet. Cependant, cela introduit un couplage fort entre sous-projets.

La méthode recommandée est plutôt de déclarer la sortie de la tâche dont on veut dépendre comme une sortie du source set main et d'effectuer une dépendance sur le projet



Exemple

build.gradle

```
task buildInfo(type: BuildInfo) {  
    version = project.version  
    outputFile = file("${buildDir}/generated-resources/build-info.properties")  
}  
  
sourceSets {  
    main {  
        output.dir(buildInfo.outputFile.parentFile, builtBy: buildInfo)  
    }  
}
```

build.gradle

```
dependencies {  
    runtime project(':producer')  
}
```




Pipeline de déploiement continue

Tests
Analyse de Code
Jenkins




Pipeline de livraison continue


Commit stage




Acceptance stage



UAT



Production





Tests



Rappels

Par défaut :


- Les classes de tests sont dans *src/test/java*
- Les ressources dans *src/test/resources*
- Les classes compilées dans *build/classes/test*

Les frameworks de test produisent les résultats des exécutions généralement

- Au format XML (*build/test-results*)
- Des rapports au format HTML (*build/reports/test*) .

2 configurations de classpath sont disponibles :
testCompile et *testRuntime*

Tâches de test





Détection des tests

Les classes de tests sont détectées comme suit :

- Toute classe étendant ***junit.framework.TestCase*** ou ***groovy.util.GroovyTestCase***
- Toute classe annotée par ***@RunWith.***
- Toute classe contenant une méthode annotée par ***@Test.*** (JUnit ou Test NG)



Dépendance *JUnit*

```
project(':repository') {  
    repositories {  
        mavenCentral()  
    }  
    dependencies {  
        compile project(':model')  
        testCompile 'junit:junit:4.11'  
    }  
}
```



Muti-framework

Si l'on veut utiliser plusieurs frameworks de test unitaires, il faut adopter une convention de nommage des tests qui par défaut est :

- JUnit: **Test.java*
- Test NG : **NGTest.java*
- Spock: **Spec.groovy*



Exemple

```
/* Les tests JUnit et Spock sont exécutés par test
* Les tests Test NG sont exécutés par la tâche testNG */
project(':repository') {
    apply plugin: 'groovy'
    repositories {
        mavenCentral()
    }
    dependencies {
        compile project(':model')
        testCompile 'junit:junit:4.11'
        testCompile 'org.testng:testng:6.8'
        testCompile 'org.codehaus.groovy:groovy:2.0.6'
        testCompile 'org.spockframework:spock-core:0.7-groovy-2.0'
    }
    task testNG(type: Test) {
        useTestNG()
    }
    test.dependsOn testNG
}
```



Options de la ligne de commande


Un gabarit du nom de la classe de test peut être fourni par la propriété système ***<taskName>.single***

```
$ gradle  
-Dtest.single=**/*Spec :repository:test
```

La propriété ***<taskName>.debug*** permet d'attacher un debugger

```
$ gradle -Dtest.debug :repository:test
```

Tâche Test





JVM

Gradle exécute les tests dans une nouvelle JVM.

On peut donc passer les options habituelles d'une JVM ou fournir des propriétés système

```
test {  
    systemProperty 'items', '20'  
    minHeapSize = '128m'  
    maxHeapSize = '256m'  
    jvmArgs '-XX:MaxPermSize=128m'  
}
```



Logger

`<<interface>>`
`TestLoggingContainer`



`<<interface>>`
`TestLogging`

```
setEvents(events: Iterable<?>)
setExceptionFormat(format: Object)
setShowExceptions(flag: boolean)
setShowStackTraces(flag: boolean)
setShowStandardStreams(flag: boolean)
```



Exemple

```
test {  
  testLogging {  
    // Afficher les System.out  
    showStandardStreams = true  
    // Toute la stack trace  
    exceptionFormat 'full'  
    // Événements de test  
    events 'started', 'passed', 'skipped', 'failed'  
  }  
}
```



Exécution parallèle

Gradle permet l'exécution des tests en parallèle.

Il suffit de préciser :

- Le nombre de JVMs à démarrer
- En option le maximum de classes de test à exécuter par processus

```
test {  
    forkEvery = 5  
    maxParallelForks =  
        Runtime.getRuntime().availableProcessors() / 2  
}
```



Callback

On peut réagir aux événements qui surviennent lors des tests :

- ***beforeSuite / afterSuite*** : Avant/après qu'une suite de test soit exécutée
- ***beforeTest / afterTest*** : Avant/après qu'une classe de test soit exécutée



Tests d'intégration

Les tests d'intégration sont généralement plus long à s'exécuter et nécessitent des systèmes externes (BD, serveur, ...)

Au niveau du build, cela implique :

- Fournir des tâches séparées pour exécuter les tests unitaires et les tests d'intégration.
- Séparer les résultats et les rapports des 2 types de test.
- Déclencher les tests d'intégration lors de tâche de cycle de vie *check*



Techniques

2 techniques peuvent facilement s'implémenter avec *Gradle* :

- Se baser sur des règles de nommage des classes de test pour différencier tests unitaires et tests d'intégration
- Définir un nouveau *sourceSet* pour les tests d'intégration. (mieux)



Convention de nommage

```
project(':repository') {
    Repositories { mavenCentral() }
    dependencies {
        compile project(':model')
        runtime 'com.h2database:h2:1.3.170'
        testCompile 'junit:junit:4.11'
    }
    test {
        exclude '**/*IntegTest.class'
        reports.html.destination = file ("$reports.html.destination/unit")
        reports.junitXml.destination = file("$reports.junitXml.destination/unit")
    }
    task integrationTest(type: Test) {
        include '**/*IntegTest.class'
        reports.html.destination = file("$reports.html.destination/integration")
        reports.junitXml.destination =
            file("$reports.junitXml.destination/integration")
    }
    check.dependsOn integrationTest
}
```




sourceSet

En ajoutant un *sourceSet*, Gradle ajoute automatiquement les tâches requises pour compiler, tester ce nouveau *sourceSet*. Les tâches sont dérivées du nom du *sourceSet*

```
sourceSets {
    integrationTest {
        java.srcDir file('src/integTest/java')
        resources.srcDir file('src/integTest/resources')
        compileClasspath = sourceSets.main.output
                           + configurations.testRuntime
        runtimeClasspath = output + compileClasspath
    }
}
task integrationTest(type: Test) {
    testClassesDir = sourceSets.integrationTest.output.classesDir
    classpath = sourceSets.integrationTest.runtimeClasspath
}
check.dependsOn integrationTest
```



Exemple démarrage serveur H2

```
class H2DatabaseStarter extends DefaultTask {  
    @Input  
    Integer tcpPort  
  
    @Input  
    Integer blockMs  
  
    @TaskAction  
    void start() {  
        new Thread(new H2Server(tcpPort)).start()  
        Thread.sleep(blockMs)  
    }  
  
    private class H2Server implements Runnable {  
        final Integer tcpPort  
  
        H2Server(Integer tcpPort) {  
            this.tcpPort = tcpPort  
        }  
  
        @Override  
        void run() {  
            org.h2.tools.Server.main('-tcp', '-tcpPort', tcpPort.toString())  
        }  
    }  
}
```



Script pour BD

```
buildscript {  
    repositories { mavenCentral() }  
    dependencies { classpath 'com.h2database:h2:1.3.170' }  
}  
ext.h2TcpPort = 9092  
ext.h2StartupBlockMs = 2000  
  
task startDatabase(type: H2DatabaseStarter) {  
    tcpPort = h2TcpPort  
    blockMs = h2StartupBlockMs  
}  
  
task stopDatabase(type: JavaExec) {  
    classpath = buildscript.configurations.classpath  
    main = 'org.h2.tools.Server'  
    args = ['-tcpShutdown', "tcp://localhost:${h2TcpPort}"]  
}  
  
task buildSchema(type: JavaExec, dependsOn: startDatabase) {  
    classpath = buildscript.configurations.classpath  
    workingDir = projectDir  
    main = 'org.h2.tools.RunScript'  
    args = ['-url', 'jdbc:h2:~/todo', '-user', 'sa', '-script', 'create-todo.sql']  
}
```



Analyse de code

L'analyse de code doit faire partie de la pipeline de déploiement continu

Les analyses peuvent également être longues à s'exécuter. C'est pourquoi, il faut disposer de tâches *Gradle* séparées au niveau du build.

Ces tâches sont généralement fournies par des plugins



Couverture des tests avec *Jacoco*

Le framework **JaCoCo** est le framework le plus habituellement répandu.

Un plugin Gradle est directement applicable par
`apply plugin: "jacoco"`

Une nouvelle tâche ***jacocoTestReport*** dépendante de *test* est disponible.

Un rapport est généré dans
\$buildDir/reports/jacoco/test. Par défaut, un rapport HTML est également généré.



Intégration Sonar

Le plugin **org.sonarqube** permet de démarrer une analyse Sonar via *Gradle*

Les propriétés de l'instance Sonar sont définies dans *gradle.properties*

```
systemProp.sonar.host.url=http://localhost:9000  
systemProp.sonar.login=<token>
```

Le plugin *org.sonarqube* est appliqué

```
plugins {  
    id "org.sonarqube" version "2.6.2"  
}
```

La tâche *sonarqube* est alors disponible, de nombreuses propriétés Sonar peuvent être affinées

```
gradle sonarqube
```



Intégration Jenkins

L'intégration Jenkins est facilité via le *wrapper* Gradle qui assure que la même version de *gradle* est installée et utilisé sur l'esclave exécutant le Job

Il est ainsi aisé :

- De faire un freestyle Job
- Ou utiliser les pipeline Jenkins (également Groovy)



Merci !!

Pour votre attention !